
CHIFFREMENT DE HILL 
                            TI-Nspire CAS 

 

 
Mots-clés : codage, décodage, clé, chiffrement, message, décalage, substitution, programme. 
 
Fichiers associés : hill.tns 
 
1. Objectifs     

Étudier la méthode de chiffrement polygraphique de Hill dans le cas bigraphique (chiffrement par paquets de 2).  
Créer des programmes pour automatiser le codage et le décodage.  

2. Présentation 

En 1929, Lester S. Hill (1891 – 1961), mathématicien américain, a l’idée de proposer un codage en groupant 
les lettres du message par bloc de n lettres.  
Nous nous proposons de coder des messages avec ce procédé.   

N.B. Les élèves familiarisés avec le calcul matriciel pourront traiter la partie 4 plutôt que la partie 3. Ceux 
qui n’ont pas encore étudié les matrices se limiteront à la partie 3.  

3. Conduite de l’activité 

Les lettres de l’alphabet seront numérotées de 0 à 25.  

A B C D E F G H I J K L M N O P Q R S T U V W X Y Z 

0 1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19 20 21 22 23 24 25 

Les calculs peuvent conduire à des résultats n  inférieurs à 0 ou supérieurs à 25. Il faut donc utiliser le reste 
de la division de n par 26 pour obtenir un nombre de 0 à 25. La lettre numérotée 26, 52, etc. est alors A 
(comme la lettre numérotée 0) et ainsi de suite.  

Dans les messages, on ne tiendra pas compte des espaces ni, quand il y en a, des accents sur les lettres.  

A. Codage du message 

1. Message : BEAUCOUP DE BRUIT.  

On groupe les lettres par deux :  BE   AU    CO     UP   DE   BR    UI    TA (complété par A).  
On numérote les lettres :  1.4  0.20  2.14  20.15  3.4  1.17  20.8  19.0  

La transformation se fait de la manière suivante :  

Si a et b sont les deux nombres, on obtient deux nouveaux nombres c et d par :  
11 5 (modulo 26)

7 4 (modulo 26)

c a b

d a b

≡ +
 ≡ +

.  

Par ce procédé, on obtient, pour le message ci-dessus :  
5.23   22.2   14.18   9.18   1.11   18.23   0.16   1.3 
Soit, en repassant aux lettres, le message chiffré suivant :  
F X W C O S J S B L S X A Q B D 

Pour un calcul non automatisé des nombres du message chiffré, voir le fichier hill.tns , Activité 1.  

2. Message : TEL EST PRIS QUI CROYAIT PRENDRE 

Message chiffré obtenu : VTLPHUQRWYQKUMXTEMBCQRFCOLSC 

B. Recherche du décodage 

1. Résolution du système 
11 5 (modulo 26)

7 4 (modulo 26)

c a b

d a b

≡ +
 ≡ +

.  

Résolvons le système par combinaisons :  
4c ≡ 44a + 20b 
5d ≡ 35a + 20b 
Soit, par différence : 4c – 5d ≡ 9a.  
La division n’est pas possible modulo 26. En revanche, comme 9 est premier avec 26, il est inversible 
modulo 26, c’est-à-dire qu’il existe un nombre i tel que 9 × i ≡ 1 (modulo 26).  
On trouve aisément :  9 × 3 = 27 ≡ 1 (mod 26). 
On obtient alors : 12c – 15d ≡ a  soit  12c + 11d ≡ a (mod 26).  

Opérons de même pour trouver b :  
7c ≡ 77a + 35b  
11d ≡ 77a + 44b  
Soit, par différence :  
–7c + 11d ≡ 9b.  
Multiplions par 3 ; on obtient alors : –21c + 33d ≡ b, soit  5c + 7d ≡ b  (modulo 26).  
On obtient donc le système :  

12 11 (modulo 26)

5 7 (modulo 26)

a c d

b c d

≡ +
 ≡ +

. 

2. Décodage 
a. On vérifie que l’on obtient bien les deux messages initiaux du A, à partir des deux messages chiffrés.  

b. Message chiffré : LPIALPNWSXNWLP. 
Message initial correspondant : LE SOLEIL BRILLE 

C. Une programmation 

Programme codage1  

Define codage1(texte)= 

Prgm 

:Local m,chiffre,long,n,i,a,b,c,d 

:""→chiffre 

:dim(texte)→long 

:If int(long/2)≠long/2 Then 

:texte&"A"→texte 

:int(long/2)+1→n 

:Else 

:long/2→n 

:EndIf 

:For i,1,n 

:  ord(mid(texte,2*i-1,1))-65→a 

:  ord(mid(texte,2*i,1))-65→b 

: mod(11a+5b,26)→c 

: mod(7a+4b,26)→d 

:  chiffre&char(c+65)→chiffre 

:  chiffre&char(d+65)→chiffre 

:EndFor 

:Disp chiffre 

:EndPrgm 

 
 

Déclaration des variables locales 

Message final vide placé dans la variable chiffre 

Longueur du texte initial 

Si le nombre de lettres est impair 

ajouter A en fin de texte  

définir le nombre n de groupes de 2 lettres 

Si le nombre de lettres est pair,  

définir le nombre n de groupes de 2 lettres 

 

De i = 1 à n 

Coder par deux nombres les deux lettres du groupe 

numéro i 

Prendre les restes des divisions par 26 des deux 

expressions 11a + 5b et 7a + 4b  

Transformer en lettres les nombres obtenus et les placer 

à droite du message chiffre 

 

Afficher le message chiffré obtenu 

Programme decodage1  

Define decodage(chiffre)= 

Prgm 

:Local m,texte,long,n,i,a,b,c,d 

:""→texte 

:dim(chiffre)→long 

:long/2→n 

:For i,1,n 

:  ord(mid(chiffre,2*i-1,1))-65→a 

:  ord(mid(chiffre,2*i,1))-65→b 

:  mod(12a+11b,26)→c 

:  mod(5a+7b,26)→d 

:  texte&char(c+65)→texte 

:  texte&char(d+65)→texte 

:EndFor 

:Disp texte 

:EndPrgm 

 
 

Déclaration des variables locales 

Message final vide placé dans la variable texte 

Longueur du texte chiffré initial 

Définir le nombre n de groupes de 2 lettres 

De i = 1 à n 

Coder par deux nombres les deux lettres du groupe 

numéro i 

Prendre les restes des divisions par 26 des deux 

expressions 12a + 11b et 5a + 7b 

Transformer en lettres les nombres obtenus et les placer 

à droite du message chiffre 

 

Afficher le message obtenu 

Le programme de décodage est moins long puisque le message chiffré comprend nécessairement un nombre 
pair de lettres. Du coup, le message obtenu peut comprendre une lettre en trop. Il n’est pas possible de savoir 
d’emblée, dans un texte chiffré, si une lettre a été ajoutée ou non. Il faudra, si besoin est, la supprimer en 
fonction de la compréhension du message obtenu. 

4. Point de vue matriciel et programmation  

Cette partie peut ne pas être traitée ou être traitée en remplacement de la partie 3.  

A. Codage du message 

1. Message : BEAUCOUP DE BRUIT. 

On groupe les lettres par deux :  BE   AU    CO     UP   DE   BR    UI    TA (complété par A).  
On numérote les lettres :  1.4  0.20  2.14  20.15  3.4  1.17  20.8  19.0  

La transformation se fait de la manière suivante :  

Si a et b sont les deux nombres, on obtient deux nouveaux nombres c et d par :  
11 5

7 4

 
 
 

*
a

b

 
 
 

 = 
x

y

 
 
 

 et   c ≡ x (modulo 26), d ≡ y (modulo 26).  

Par ce procédé, on obtient, pour le message ci-dessus :  
5.23   22.2   14.18   9.18   1.11   18.23   0.16   1.3 
Soit, en repassant aux lettres, le message chiffré suivant :  
F X W C O S J S B L S X A Q B D 

2. Message : TEL EST PRIS QUI CROYAIT PRENDRE 

Message chiffré obtenu : VTLPHUQRWYQKUMXTEMBCQRFCOLSC 

B. Recherche du décodage 

1. Résolution du système 

La comatrice de M = 
11 5

7 4

 
 
 

est  : 
4 7

5 11

− 
 − 

.     

La transposée de la matrice obtenue est : M’  = 
4 5

7 11

− 
 − 

. 

On recherche le déterminant de M et on l’inverse modulo 26. Ici, le déterminant est 9 et son inverse modulo 
26 est 3 puisque 9 × 3 ≡ 1 (mod 26).  

On multiplie la matrice M’  par cet inverse : 3 × 
4 5

7 11

− 
 − 

 = 
12 15

21 33

− 
 − 

. 

On détermine les restes de la division par 26 des termes de la matrice et l’on obtient ainsi la matrice de 

décodage : 
12 11

5 7

 
 
 

.  

2. Décodage 
a. On vérifie que l’on obtient bien les deux messages initiaux du A, à partir des deux messages chiffrés.  

b. Message chiffré : LPIALPNWSXNWLP. 

Message initial correspondant : LE SOLEIL BRILLE 

C. Une programmation 

Programme codage  

Define codage(texte)= 

Prgm 

:Local m,chiffre,long,n,i,a,b,c,d,p 

:[11,5;7,4]→m 

:""→chiffre 

:dim(texte)→long 

:If int(long/2)≠long/2 Then 

:texte&"A"→texte 

:int(long/2)+1→n 

:Else 

:long/2→n 

:EndIf 

:For i,1,n 

:  ord(mid(texte,2*i-1,1))-65→a 

:  ord(mid(texte,2*i,1))-65→b 

:  m*[a;b]→p 

: mod(p[1,1],26)→c 

: mod(p[2,1],26)→d 

:  chiffre&char(c+65)→chiffre 

:  chiffre&char(d+65)→chiffre 

:EndFor 

:Disp chiffre 

:EndPrgm 

 
 

Déclaration des variables locales 

Définition de la matrice de codage 

Message final vide placé dans la variable chiffre 

Longueur du texte initial 

Si le nombre de lettres est impair 

ajouter A en fin de texte  

définir le nombre n de groupes de 2 lettres 

Si le nombre de lettres est pair,  

définir le nombre n de groupes de 2 lettres 

 

De i = 1 à n 

Coder par deux nombres les deux lettres du groupe 

numéro i 

Appliquer au vecteur colonne obtenu la matrice m 

Prendre les restes des divisions par 26 des deux termes 

du vecteur colonne p  

Transformer en lettres les nombres obtenus et les placer 

à droite du message chiffre 

 

Afficher le message chiffré obtenu 

Le programme ci-dessus tient compte du nombre pair ou impair de lettres du message à chiffrer. 

En revanche, page suivante, le programme de décodage sera moins long puisque le message chiffré 
comprend nécessairement un nombre pair de lettres.  

Du coup, le message décodé obtenu peut comprendre une lettre en trop. Il n’est pas possible de savoir 
d’emblée, dans un texte chiffré par la méthode de Hill, si une lettre a été ajoutée ou non. Il faudra, si besoin 
est, la supprimer en fonction de la compréhension du message obtenu.  

Par exemple, avec le programme de décodage (comme en décodant « à la main »), le décodage du message 
chiffré F X W C O S J S B L S X A Q B D donne BEAUCOUPDEBRUITA, dont il convient de supprimer le 
A final.  

Programme decodage  

Define decodage(chiffre)= 

Prgm 

:Local m,texte,long,n,i,a,b,c,d 

:[12,11;5,7]→m 

:""→texte 

:dim(chiffre)→long 

:long/2→n 

:For i,1,n 

:  ord(mid(chiffre,2*i-1,1))-65→a 

:  ord(mid(chiffre,2*i,1))-65→b 

:  m*[a;b]→p 

:  mod(p[1,1],26)→c 

:  mod(p[2,1],26)→d 

:  texte&char(c+65)→texte 

:  texte&char(d+65)→texte 

:EndFor 

:Disp texte 

:EndPrgm 

 
 

Déclaration des variables locales 

Définition de la matrice de décodage 

Message final vide placé dans la variable texte 

Longueur du texte chiffré initial 

Définir le nombre n de groupes de 2 lettres 

De i = 1 à n 

Coder par deux nombres les deux lettres du groupe 

numéro i 

Appliquer au vecteur colonne obtenu la matrice m 

Prendre les restes des divisions par 26 des deux termes 

du vecteur colonne p 

Transformer en lettres les nombres obtenus et les placer 

à droite du message chiffre 

 

Afficher le message obtenu 

   Le programme de recherche de la matrice de décodage est calqué sur la partie B. 1. précédente.   

Programme recherche  

Define recherche(m11,m12,m21,m22)= 

Prgm 

: Local m,n,i,k 
:[m11,m12;m21,m22]→m 

:det(m)→n 

:If gcd(n,26)=1 Then 

:For i,0,25 

:If mod(n*i,26)=1:i→k 

:EndFor 

:©comatrice 

:[m22,−m21;−m12,m11]→m 

:©transposee de la comatrice 

:m
T
→m 

:mod(k*m,26)→m 

:Disp m 

:Else 

:Disp "changer de matrice de codage, car 

pgcd(det,26)≠1" 

:EndIf 

:EndPrgm 

 
 

Déclaration des variables locales 

Définition de la matrice m 

Calcul du déterminant n 

Vérifier que n et 26 sont premiers entre eux 

Recherche de l’inverse k, modulo 26, du déterminant n 

de la matrice m 

 

Remplacement de la matrice donnée par sa comatrice 

 

Recherche de la transposée de la comatrice 

 

Calcul de la matrice finale (reste de la division de k * m 

par 26), puis affichage de la matrice de décodage 

Sinon, (si pgcd(n,26) ≠ 1), 

afficher la nécessité de changer de matrice de codage  

 

Une source : 
Robert Noirfalise, Arithmétique et cryptographie, IREM de Clermont-Ferrand,    
http://wwwmaths.univ-bpclermont.fr/irem/ressources/lycee/arithmetique/pdf/crypto.pdf 

 

 

 


